Q:

Listview - populate second column with corresponding items from database

Fist column of listview is populated from database (database column name KONTO - only values that starts with 2020-), second column of the listview should be populate from corresponding items to KONTO from same database column NAZIV into second column of the listview.

Description is in column NAZIV in databsase
I've been experimenting with this all day and didn't have any success. 
Here is the code I have so far :
        ListView1.View = System.Windows.Forms.View.Details
    ListView1.Columns.Add("COL1", 100, HorizontalAlignment.Left) 'KONTO
    ListView1.Columns.Add("COL2", 160, HorizontalAlignment.Left) 'NAZIV

    Dim FilePath As String = "W:\GLAVNI\KOR14\"
    Dim DBF_File As String = "MATIKGL"
    Dim ColName As String = "KONTO"
    'Dim naz As String
    Using con As New OleDbConnection("Provider=Microsoft.Jet.OLEDB.4.0;Data Source=" & FilePath & _
    " ;Extended Properties=dBASE IV")
        con.Open()
        Using cmd As New OleDbCommand("SELECT * FROM MATIKGL ORDER BY KONTO, NAZIV", con)
            Using reader As OleDbDataReader = cmd.ExecuteReader()
                If reader.HasRows Then
                    While (reader.Read())
                        Me.ListView1.Items.Add(reader("KONTO"))

                        'LOOP BELOW SELECTS ALL ITEMS THAT STARTS WITH 2020-
                        For i = 0 To ListView1.Items.Count - 1
                            If ListView1.Items(i).ToString.Contains("2020-") Then

                            Else
                                ListView1.Items.Remove(ListView1.Items(i))
                            End If
                        Next

                    End While

                Else

                End If
            End Using
        End Using
        con.Close()
    End Using

Thanks.
UPDATE
To be more clear, here is the screen from database.

Only items starting with 2020- (1) should populate first column of listview, and second column should  be populated with "2020-" description from column NAZIV (2).

A:

Each LV Item has a collection of SubItems associated with it.  The subItems show as columns in the Details view:
Dim lvi As ListViewItem              ' scratch var for adding items
Dim tmp As String

Using cmd As New OleDbCommand("SELECT * FROM MATIKGL ORDER BY KONTO, NAZIV", con)
        Using reader As OleDbDataReader = cmd.ExecuteReader()
            If reader.HasRows Then
                While (reader.Read())

                    ' Db rdr returns Object, cast to string
                    tmp = reader.Item("KONTO").ToString

                    ' no need to loop - just use the scratch vars
                    ' only adding if it contains 2020
                    If tmp.Contains("2020-") Then
                        lvi = New ListViewItem
                        lvi.Text = tmp

                        ' add the sub item
                        lvi.SubItems.Add(reader("NAZIV").toString)
                        ListView1.Items.Add(lvi)
                    End If
               End While
            End If
        End Using

you can get rid of the tmp var using reader.Item("KONTO").ToString.Contains("2020")...the above is for clarity

